


# Lab 1 Multiples of 2 or 3 Detector

## Example

To help you understand Verilog, we provide an example below.



Here is a 4-bit module named "happy\_verilog", and its function detecting prime numbers between 0 and 15.

We will use it to help you understand three different ways of describing a circuit.


- Prime numbers: 2, 3, 5, 7, 11, 13, 17, 19, ...

Therefore, here we need to find 2, 3, 5, 7, 11, 13.

## Description

1. Three ways to describe the circuit.

There are three ways to describe the circuit, “gate level”, “dataflow”, and “behavior”. We will talk more about it in the following parts.




2. K-map

cd \ ab	00	01	11	10
00	0	1	3	2
01	4	5	7	6
11	12	13	15	14
10	8	9	11	10

### 3. Structure

Simplify the K-map, you will get  $F(a, b, c, d) = a'b'c + a'cd + b'cd + bc'd$



a, b, c, d are in[3], in[2], in[1], in[0] in the circuit.

### 4. Implement in Verilog

#### (1) Gate Level Description

```
1  module happy_verilog_G(in, out);
2
3      parameter n = 4;
4
5      //IO port declaration
6      input [n - 1: 0]in;//Input must be wire.
7      output out;//Output can be wire or reg, default type is wire.
8
9      //Save the output values from different gates.
10     wire not_a, not_b, not_c;
11     wire and0, and1, and2, and3;
12
13     //Declare a gate.
14     //<gate><gate name>(output,input1,input2,...);
15     //Don't use duplicate <gate name>
16     not not_gate0(not_a, in[3]);
17     not not_gate1(not_b, in[2]);
18     not not_gate2(not_c, in[1]);
19
20     and and_gate0(and0, not_a, not_b, in[1]);
21     and and_gate1(and1, not_a, in[1], in[0]);
22     and and_gate2(and2, not_b, in[1], in[0]);
23     and and_gate3(and3, in[2], not_c, in[0]);
24
25     or or_gate(out, and0, and1, and2, and3);
26
27 endmodule
```

## (2) Dataflow Description

```
1  module happy_verilog_D(in, out);
2
3      parameter n = 4;
4
5      input [n - 1: 0]in;
6      output out;
7
8      //Be careful, "!" and "~" are different in Verilog.
9      assign out = (!in[3] & !in[2] & in[1])
10                  | (!in[3] & in[1] & in[0])
11                  | (!in[2] & in[1] & in[0])
12                  | (in[2] & !in[1] & in[0]);
13
14  endmodule
```

## (3) Behavior Description

```
1  module happy_verilog_B(in, out);
2
3      parameter n = 4;
4
5      input [n - 1: 0]in;
6      output out;
7      reg out; //It must be reg type because an always block is used to set it.
8
9      always@(*)begin
10         case(in)
11             2, 3, 5, 7, 11, 13:begin
12                 out = 1'b1;
13             end
14             default:begin
15                 out = 1'b0;
16             end
17         endcase
18     end
19
20     /*
21     or can write as
22     always@(*)begin
23         out = 1'b0;
24         case(in)
25             2, 3, 5, 7, 11, 13:begin
26                 out=1'b1;
27             end
28         endcase
29     end
30     */
31
32  endmodule
```

## 5. Testbench

```
1  module happy_verilog_tb;
2      parameter delay = 5;
3
4      wire out_G, out_D, out_B;
5      reg [3: 0]in;
6      integer i;
7
8      wire is;
9      assign is = in == 2 || in == 3 || in == 5 || in == 7 || in == 11 || in == 13;
10
11     initial begin
12         //Give initial value.
13         in = 4'd0;
14         for(i = 0; i < 16; i = i + 1)begin
15             #delay;
16             $display("time = %4d, in = %b, out_G = %b, out_D = %b, out_B = %b",
17                 $time, in, out_G, out_D, out_B);
18             if((out_G === 1'bx | out_D === 1'bx | out_B === 1'bx |
19                 out_G === 1'bz | out_D === 1'bz | out_B === 1'bz) ||
20                 ( is && (out_G & out_D & out_B) == 0) ||
21                 (!is && (out_G | out_D | out_B) == 1))
22                 begin
23                     $display("You got wrong answer!!");
24                     $finish;
25                 end
26             in = in + 4'd1;
27         end
28         $display("Congratulations!!");
29         $finish;
30     end
31
32     //Connect with the "gate level" module.
33     happy_verilog_G hvg( .in(in), .out(out_G) );
34
35     //Connect with the "dataflow" module.
36     happy_verilog_D hvd( .in(in), .out(out_D) );
37
38     //Connect with the "behavior" module.
39     happy_verilog_B hvb( .in(in), .out(out_B) );
40
41 endmodule
```

Write and run the code by yourself, it's a great chance to practice.

# Use NC-Verilog to simulate and verify

## How to execute

Use command, “ncverilog”, to execute testbench and modules.

Here is an example.

Assume testbench is written in a file called “happy\_verilog\_tb.v”

And all three modules are written in a file called “happy\_verilog.v”

Then our command would be:

```
ncverilog happy_verilog_tb.v happy_verilog.v
```

```
~/lab1]$ ncverilog happy_verilog_tb.v happy_verilog.v
```

## Result

Right answer:

```
time = 5, in = 0000, out_G = 0, out_D = 0, out_B = 0
time = 10, in = 0001, out_G = 0, out_D = 0, out_B = 0
time = 15, in = 0010, out_G = 1, out_D = 1, out_B = 1
time = 20, in = 0011, out_G = 1, out_D = 1, out_B = 1
time = 25, in = 0100, out_G = 0, out_D = 0, out_B = 0
time = 30, in = 0101, out_G = 1, out_D = 1, out_B = 1
time = 35, in = 0110, out_G = 0, out_D = 0, out_B = 0
time = 40, in = 0111, out_G = 1, out_D = 1, out_B = 1
time = 45, in = 1000, out_G = 0, out_D = 0, out_B = 0
time = 50, in = 1001, out_G = 0, out_D = 0, out_B = 0
time = 55, in = 1010, out_G = 0, out_D = 0, out_B = 0
time = 60, in = 1011, out_G = 1, out_D = 1, out_B = 1
time = 65, in = 1100, out_G = 0, out_D = 0, out_B = 0
time = 70, in = 1101, out_G = 1, out_D = 1, out_B = 1
time = 75, in = 1110, out_G = 0, out_D = 0, out_B = 0
time = 80, in = 1111, out_G = 0, out_D = 0, out_B = 0
Congratulations!!
```

(All testcases are correct!)

Wrong answer:

```
time = 5, in = 0000, out_G = 0, out_D = 0, out_B = 0
time = 10, in = 0001, out_G = 0, out_D = 0, out_B = 0
time = 15, in = 0010, out_G = 1, out_D = 1, out_B = 1
time = 20, in = 0011, out_G = 1, out_D = 1, out_B = 1
time = 25, in = 0100, out_G = 0, out_D = 0, out_B = 0
time = 30, in = 0101, out_G = 1, out_D = 1, out_B = 1
time = 35, in = 0110, out_G = 0, out_D = 0, out_B = 0
time = 40, in = 0111, out_G = 1, out_D = 1, out_B = 1
time = 45, in = 1000, out_G = 0, out_D = 0, out_B = 0
time = 50, in = 1001, out_G = 0, out_D = 0, out_B = 0
time = 55, in = 1010, out_G = 0, out_D = 0, out_B = 0
time = 60, in = 1011, out_G = 1, out_D = 1, out_B = 0
You got wrong answer!!
```

(The last testcase in the image is wrong!)

# Assignment

Design a 4-bit input circuit that can detect multiples of 2 or 3, and simplify it using Karnaugh maps. Then, write Verilog code for the circuit using gate-level, dataflow, and behavioral descriptions.

- multiples of 2 or 3: 0, 2, 3, 4, 6, 8, 9, ...etc.
- Gate Level Description
  - module name: MultiplesDetector\_G
  - input name: in
  - output name: out
- Dataflow Description
  - module name: MultiplesDetector\_D
  - input name: in
  - output name: out
- Behavior Description
  - module name: MultiplesDetector\_B
  - input name: in
  - output name: out

Please write three description in different modules and put all of them in one file, “MultiplesDetector.v”

Write testbench in a file called “MultiplesDetector\_tb.v”

Follow the example above, design your testbench and test your own Multiples Detector.

## Files to upload

1. MultiplesDetector.v
2. {studentID}\_report.pdf (For example, 111062000\_report.pdf)
  - (1) K-map, circuit diagram, and execution result (screen shot).
  - (2) The problem you faced and how you deal with it.
  - (3) What have you learned from this lab?

Please upload these two files on the eeclass system.

Be careful of the file name, module name, and whatever listed above.

Wrong format will result in the failure(0 points) of your assignment, please double check before uploading it.

## Deadline

- 2023/04/26 23:59

Upload the files before the deadline!

Late assignment will get a 10% penalty per day, and no assignment will be accepted after 3 days from the given due date.

Do not plagiarize!